

回答 投稿者: hatena

投稿日時: 14/08/24 17:22:30



合計探索問題、みなさんの頭脳に挑戦!!^^; ---- ご意見・ご感想・ご主張 BBS ----

<http://www.accessclub.jp/bbs7/0002/bbs989.html>

の私のコードをもとに今回の案件用に改良してみました。

A1:B50に下記のデータを貼り付け。(タブ区切りで入力してあるのでコピーして貼り付けられます。)

C1 に探索する合計値を入力します。

5988	うう	88792
2994	ええ	
1245	うう	
3296	ええ	
19777	うう	
1497	うう	
14823	うう	
13177	うう	
37885	ええ	
5988	うう	
6290	ええ	
6038	うう	
22653	ええ	
28474	うう	
29564	ええ	
26871	うう	
23844	おお	
4366	ええ	
4101	おお	
14116	うう	
7037	ええ	
17500	うう	
24062	おお	
23644	うう	
17717	ええ	
25162	うう	
9461	おお	
19788	うう	
29762	ええ	
25099	おお	
28935	うう	
1011	おお	
4655	うう	
22234	ええ	
9589	うう	
30377	おお	
10081	うう	
2887	おお	
24336	うう	
3517	ええ	
16020	おお	
6494	おお	
16745	おお	
24100	ええ	
28340	ええ	
24825	おお	
13382	ええ	
6801	ええ	
19893	ええ	
28700	おお	

```
Option Explicit
Option Base 1 '配列の添字の最小値を1に変更
Const MaxCnt = 100 '最大取得件数

Private Cnt As Long '検索件数
Private rowCnt As Long 'データ件数

Public Sub DoSearch()
    Dim K, KName
    Dim a()
    Dim b()
    Dim i As Long, c As Long

    Range("D1", ActiveCell.SpecialCells(xlLastCell)).ClearContents
    Range("C3").Value = ""

    rowCnt = Range("B" & Rows.Count).End(xlUp).Row
    Cnt = 0
    a = Range("A1", "B" & rowCnt)

    Set K = CreateObject("Scripting.Dictionary") '会社名と会社名毎件数を格納
    For i = 1 To rowCnt
        KName = Cells(i, 2).Value
        If Not K.Exists(KName) Then
            K.Add KName, 1
        Else
            K(KName) = K(KName) + 1
        End If
    Next

    For Each KName In K
        ReDim b(K(KName), 2)
        c = 0
        For i = 1 To UBound(a, 1)
            If a(i, 2) = KName Then
                c = c + 1
                b(c, 1) = a(i, 1)
                b(c, 2) = i
            End If
        Next

        If Not SearchSum(Cells(1, 3).Value, b) Then
            Range("C3").Value = "最大件数を越えたので中止しました。"
            Exit For
        End If
    Next

    Range("C2").Value = Cnt & "件"
End Sub
```

```
'合計探索関数
'引数 TargetSum 探索合計
'      aryData() データ と 行位置 の二次元配列
Public Function SearchSum( _
    TargetSum As Long, _
    aryData() _
) As Boolean

    Dim totalSum As Long      '総合計
    Dim flgRev As Boolean
    Dim rsLast As Long       'データ件数
    Dim ary() As Long        '組み合わせ配列 レコード位置
    Dim arySum() As Long     '組み合わせ配列 合計
    Dim curSum As Long       'カレント合計金額
    Dim AP As Long          'データ配列のカレント位置
    Dim Pos As Long         '組み合わせ配列内のカレント位置
    Dim i As Long

    rsLast = UBound(aryData, 1)
    AP = 1: Pos = 1

    QuickSort2 aryData, LBound(aryData), rsLast

    For i = LBound(aryData, 1) To rsLast
        totalSum = totalSum + aryData(i, 1)
    Next

    If totalSum = TargetSum Then
        Cnt = Cnt + 1
        AllSelect
        SearchSum = True
        Exit Function
    End If
    If totalSum < TargetSum Then
        SearchSum = True
        Exit Function
    End If

    If totalSum / 2 < TargetSum Then
        flgRev = True
        TargetSum = totalSum - TargetSum
    End If

    ReDim ary(rsLast)
    ReDim arySum(0 To rsLast)

    Do
        ary(Pos) = AP
        curSum = curSum + aryData(AP, 1)
        arySum(Pos) = curSum
        If curSum = TargetSum Then
            Cnt = Cnt + 1
            If flgRev Then ShowaryRev ary, aryData Else Showary ary, aryData
            If Cnt >= MaxCnt Then Exit Function
            If Pos = 1 Then Exit Do
            ary(Pos) = 0
            Pos = Pos - 1
            curSum = arySum(Pos - 1)
            AP = ary(Pos) + 1
        ElseIf AP = rsLast Or curSum > TargetSum Then
            If Pos = 1 Then Exit Do
            ary(Pos) = 0
            Pos = Pos - 1
            curSum = arySum(Pos - 1)
            AP = ary(Pos) + 1
        ElseIf curSum + aryData(AP, 1) > TargetSum Then
            curSum = arySum(Pos - 1)
            AP = ary(Pos) + 1
        Else
            Pos = Pos + 1
            AP = AP + 1
        End If
    Loop

    SearchSum = True
End Function
```

```
Private Sub AllSelect()  
    Dim i As Integer  
  
    For i = 1 To rowCnt  
        Cells(i, Cnt + 3).Value = "〇"  
    Next  
End Sub  
  
Private Sub Showary(ary() As Long, aryData())  
    Dim i As Integer  
  
    For i = 1 To UBound(ary)  
        If ary(i) = 0 Then Exit For  
        Cells(aryData(ary(i), 2), Cnt + 3).Value = "〇"  
    Next  
End Sub  
  
Private Sub ShowaryRev(ary() As Long, aryData())  
    Dim i As Integer, s As String  
  
    AllSelect  
  
    For i = 1 To UBound(ary)  
        If ary(i) = 0 Then Exit For  
        Cells(aryData(ary(i), 2), Cnt + 3).Value = ""  
    Next  
End Sub  
  
'二次元配列ソート  
Private Sub QuickSort2(a(), L As Long, R As Long)  
    Dim s, t, i As Long, j As Long  
  
    If L < R Then  
        s = a((L + R) \ 2, 1)  
        i = L - 1: j = R + 1  
        Do While 1  
            Do  
                i = i + 1  
            Loop While a(i, 1) < s  
            Do  
                j = j - 1  
            Loop While a(j, 1) > s  
            If i >= j Then Exit Do  
            t = a(i, 1): a(i, 1) = a(j, 1): a(j, 1) = t  
            t = a(i, 2): a(i, 2) = a(j, 2): a(j, 2) = t  
        Loop  
  
        QuickSort2 a, L, i - 1  
        QuickSort2 a, j + 1, R  
    End If  
End Sub
```

DoSearch プロシーダを実行してください。

全50件、一社当たり最大20件のこのデータで、私の環境で 0.05秒以下で結果が出ました。

#### 実行結果

5988	うう	88792	○	○	
2994	ええ	4件			
1245	うう		○	○	○
3296	ええ				○
19777	うう			○	○
1497	うう				
14823	うう				
13177	うう			○	○
37885	ええ				
5988	うう		○	○	
6290	ええ				
6038	うう		○		
22653	ええ				○
28474	うう				
29564	ええ				○
26871	うう		○		
23844	おお				
4366	ええ				
4101	おお				
14116	うう				
7037	ええ				
17500	うう		○		
24062	おお				
23644	うう				
17717	ええ				
25162	うう		○		
9461	おお				
19788	うう				
29762	ええ				○
25099	おお				
28935	うう			○	○
1011	おお				
4655	うう				
22234	ええ				
9589	うう			○	○
30377	おお				
10081	うう			○	○
2887	おお				
24336	うう				
3517	ええ				○
16020	おお				
6494	おお				
16745	おお				
24100	ええ				
28340	ええ				
24825	おお				
13382	ええ				
6801	ええ				
19893	ええ				
28700	おお				

一社当たり最大30件に増やして試してみましたが、1秒以内で終了しました。

ただ、結果件数が75件になりましたので、ここから目的のものを探すのが難しいですね。

**回答**

投稿者: hatena

投稿日時: 14/08/25 02:53:12



上記の解説とサンプルをブログにアップしましたので、興味があればご参考に。

VBAで組み合わせ合計探索 - hatena chips

<http://hatenachips.blog34.fc2.com/blog-entry-430.html>